
Qbasic-Programm zur Berechnung der Laserpositionen an den Seiten und zur Fehleranalyse

DECLARE SUB MessFehlerSimulieren ()

DECLARE SUB SchnittpunkteDrehachseBerechnen ()

DECLARE SUB NormalVektorenBestimmen ()

DECLARE SUB ErstelleGleichungssystem ()

DECLARE SUB ParameterBerechnen ()

DECLARE SUB KreisMittelpunktBerechnen ()

DECLARE SUB MittelpunktBestimmen ()

DECLARE SUB GeradeBestimmen ()

DECLARE SUB MittelsenkrechteBestimmen ()

DECLARE SUB EbeneBerechnen ()
Deklaration der Subprozeduren

CONST BildBreite = 24

CONST Bildhoehe = 18

CONST AufloesungBreite = 4000

CONST Aufloesunghoehe = 3000
Konstanten für Fehleranalyse

TYPE Ebenetyp

  a AS DOUBLE

  b AS DOUBLE

  c AS DOUBLE

END TYPE

TYPE PunktTyp

  x AS DOUBLE

  y AS DOUBLE

  z AS DOUBLE

END TYPE

DIM SHARED zu AS PunktTyp

DIM SHARED halb AS PunktTyp

DIM SHARED offen AS PunktTyp

DIM SHARED Ebene AS Ebenetyp

DIM SHARED RVZuHalboffen AS PunktTyp

DIM SHARED RVZuOffen AS PunktTyp

DIM SHARED MPZuHalboffen AS PunktTyp

DIM SHARED MPZuOffen AS PunktTyp

DIM SHARED NVZuHalboffen AS PunktTyp

DIM SHARED NVZuOffen AS PunktTyp

DIM SHARED MPDZuHalboffen AS DOUBLE

DIM SHARED MPDZuOffen AS DOUBLE

DIM SHARED NVQuadratZuHalboffen AS DOUBLE

DIM SHARED NVQuadratZuOffen AS DOUBLE

DIM SHARED MPDNVZuHalboffen AS DOUBLE

DIM SHARED MPDNVZuOffen AS DOUBLE

DIM SHARED NVZuHalboffenXZuOffen AS DOUBLE

DIM SHARED Parameter AS DOUBLE

DIM SHARED mpkreis AS PunktTyp

DIM SHARED SPLinks AS PunktTyp

DIM SHARED SPRechts AS PunktTyp

CLS 
Deklarationen der Variablen  und Typen

INPUT "Geben Sie den x Wert bei geschlossenem Mund ein: "; zu.x

INPUT "Geben Sie den y Wert bei geschlossenem Mund ein: "; zu.y

INPUT "Geben Sie den z Wert bei geschlossenem Mund ein: "; zu.z

INPUT "Geben Sie den x Wert bei halboffenem Mund ein: "; halb.x

INPUT "Geben Sie den y Wert bei halboffenem Mund ein: "; halb.y

INPUT "Geben Sie den z Wert bei halboffenem Mund ein: "; halb.z

INPUT "Geben Sie den x Wert bei offenem Mund ein: "; offen.x

INPUT "Geben Sie den y Wert bei offenem Mund ein: "; offen.y

INPUT "Geben Sie den z Wert bei offenem Mund ein: "; offen.z
Einlesen der Variablen

MessFehlerSimulieren

EbeneBerechnen

GeradeBestimmen

MittelpunktBestimmen

NormalVektorenBestimmen

ErstelleGleichungssystem

ParameterBerechnen

KreisMittelpunktBerechnen

SchnittpunkteDrehachseBerechnen
Ausführen der Sub-Prozeduren



PRINT

PRINT "Die Koordinaten des Lasers auf der rechten Seite sind folgende: "

PRINT "x = "; SPRechts.x

PRINT "y = "; SPRechts.y

PRINT "z = "; SPRechts.z

PRINT "Die Koordinaten des Lasers auf der linken Seite sind folgende: "

PRINT "x = "; SPLinks.x

PRINT "y = "; SPLinks.y

PRINT "z = "; SPLinks.z

PRINT

PRINT mpkreis.x

PRINT mpkreis.y

PRINT mpkreis.z

END
Ausgabe der Ergebnisse



Subprozeduren

SUB MessFehlerSimulieren

  Zu.x = zu.x + .5 * (BildBreite / AufloesungBreite)

  zu.y = zu.y - .5 * (BildBreite / AufloesungBreite)

  zu.z = zu.z - .5 * (Bildhoehe / Aufloesunghoehe)

  halb.x = halb.x - .5 * (BildBreite / AufloesungBreite)

  halb.y = halb.y + .5 * (BildBreite / AufloesungBreite)

  halb.z = halb.z + .5 * (Bildhoehe / Aufloesunghoehe)

  offen.x = offen.x + .5 * (BildBreite / AufloesungBreite)

  offen.y = offen.y - .5 * (BildBreite / AufloesungBreite)

  offen.z = offen.z - .5 * (Bildhoehe / Aufloesunghoehe)

END SUB
Es wird untersucht, wie sich die Endergebnisse bei verschiedenen Auflösungen und verschiedenen Bildgrößen verändern, wenn man davon ausgeht, dass bei der Bildverarbeitung eine durchschnittliche Messabweichung von einem halben Pixel entstehen kann

SUB EbeneBerechnen

  Ebene.a = (zu.y * (halb.z - offen.z) + halb.y * (offen.z - zu.z) + offen.y * (zu.z - halb.z)) / (zu.x * (halb.y * offen.z - offen.y * halb.z) + halb.x * (offen.y * zu.z - zu.y * offen.z) + offen.x * (zu.y * halb.z - halb.y * zu.z))

  Ebene.b = -(zu.x * (halb.z - offen.z) + halb.x * (offen.z - zu.z) + offen.x * (zu.z - halb.z)) / (zu.x * (halb.y * offen.z - offen.y * halb.z) + halb.x * (offen.y * zu.z - zu.y * offen.z) + offen.x * (zu.y * halb.z - halb.y * zu.z))

  Ebene.c = (zu.x * (halb.y - offen.y) + halb.x * (offen.y - zu.y) + offen.x * (zu.y - halb.y)) / (zu.x * (halb.y * offen.z - offen.y * halb.z) + halb.x * (offen.y * zu.z - zu.y * offen.z) + offen.x * (zu.y * halb.z - halb.y * zu.z))

END SUB
Bei der Berechnung der Koeffizienten der Ebenen wurden die Lösungen für a, b und c nach folgendem Gleichungssystem berechnet und d = 1 gesetzt:


[image: image1.wmf]d

cz

by

ax

d

cz

by

ax

d

cz

by

ax

=

+

+

=

+

+

=

+

+

3

3

3

2

2

2

1

1

1




SUB GeradeBestimmen

  RVZuHalboffen.x = (halb.x - zu.x)

  RVZuHalboffen.y = (halb.y - zu.y)

  RVZuHalboffen.z = (halb.z - zu.z)

  RVZuOffen.x = (offen.x - zu.x)

  RVZuOffen.y = (offen.y - zu.y)

  RVZuOffen.z = (offen.z - zu.z)

END SUB
Bestimmung der Richtungsvektoren der Geraden aus der Differenz der Messpunkte P1P2 und P2P3.

SUB MittelpunktBestimmen

  MPZuHalboffen.x = .5 * (zu.x + halb.x)

  MPZuHalboffen.y = .5 * (zu.y + halb.y)

  MPZuHalboffen.z = .5 * (zu.z + halb.z)

  MPZuOffen.x = .5 * (zu.x + offen.x)

  MPZuOffen.y = .5 * (zu.y + offen.y)

  MPZuOffen.z = .5 * (zu.z + offen.z)

END SUB
Bestimmung des Mittelpunktes nach der Formel:


[image: image2.wmf](

)

2

1

2

1

P

P

M

+

=



SUB NormalVektorenBestimmen

  NVZuHalboffen.x = (Ebene.b * RVZuHalboffen.z - RVZuHalboffen.y * Ebene.c) / (Ebene.a * RVZuHalboffen.y - RVZuHalboffen.x * Ebene.b)

  NVZuHalboffen.y = (RVZuHalboffen.x * Ebene.c - Ebene.a * RVZuHalboffen.z) / (Ebene.a * RVZuHalboffen.y - RVZuHalboffen.x * Ebene.b)

  NVZuHalboffen.z = 1

  NVZuOffen.x = (Ebene.b * RVZuOffen.z - RVZuOffen.y * Ebene.c) / (Ebene.a * RVZuOffen.y - RVZuOffen.x * Ebene.b)

  NVZuOffen.y = (RVZuOffen.x * Ebene.c - Ebene.a * RVZuOffen.z) / (Ebene.a * RVZuOffen.y - RVZuOffen.x * Ebene.b)

  NVZuOffen.z = 1

END SUB
Für die Berechnung des Richtungsvektors der Mittlesenkrechten benötigen wir die Lösungen des Gleichungssystems

wobei a, b und c die Koeffzienten der Ebene, (, ( und ( die Komponenten des Richtungsvektors der Geraden durch die Messpunkte sind und z = 1 gesetzt werden kann.

SUB ErstelleGleichungssystem

  MPDNVZuHalboffen = (MPZuHalboffen.x - MPZuOffen.x) * NVZuHalboffen.x + (MPZuHalboffen.y - MPZuOffen.y) * NVZuHalboffen.y + (MPZuHalboffen.z - MPZuOffen.z) * NVZuHalboffen.z

  MPDNVZuOffen = (MPZuHalboffen.x - MPZuOffen.x) * NVZuOffen.x + (MPZuHalboffen.y - MPZuOffen.y) * NVZuOffen.y + (MPZuHalboffen.z - MPZuOffen.z) * NVZuOffen.z

  NVQuadratZuHalboffen = NVZuHalboffen.x ^ 2 + NVZuHalboffen.y ^ 2 + NVZuHalboffen.z ^ 2

  NVQuadratZuOffen = NVZuOffen.x ^ 2 + NVZuOffen.y ^ 2 + NVZuOffen.z ^ 2

  NVZuHalboffenXZuOffen = NVZuHalboffen.x * NVZuOffen.x + NVZuHalboffen.y * NVZuOffen.y + NVZuHalboffen.z * NVZuOffen.z

END SUB
Ausgehend von den Geradengleichungen der beiden Mittelsenkrechten kann der Mittelpunkt des Kreises berechnet werden. Dies entspricht der Auffindung einer Lösung für ein Gleichungssystem mit 6 Gleichungen und 5 Unbekannten. Derive ist nicht in der Lage eine allgemeine Formel für die Lösung in einem solchen Fall anzugeben und deshalb mußten wir das Problem umformulieren.

Die Punkt-Richtungsform der Mittelsenkrechte

in der die Punkte M1 und M2, sowie die Richtungsvektoren bekannt sind, wurde nach dem Parameter  

 aufgelöst, mit:

SUB ParameterBerechnen

  Parameter = (NVQuadratZuOffen * MPDNVZuHalboffen – NVZuHalboffenXZuOffen * MPDNVZuOffen) / (NVZuHalboffenXZuOffen ^ 2 - NVQuadratZuHalboffen * NVQuadratZuOffen)

END SUB


SUB KreisMittelpunktBerechnen

  mpkreis.x = MPZuHalboffen.x + Parameter * NVZuHalboffen.x

  mpkreis.y = MPZuHalboffen.y + Parameter * NVZuHalboffen.y

  mpkreis.z = MPZuHalboffen.z + Parameter * NVZuHalboffen.z

END SUB
Nachdem der Parameter berechnet worden ist, wird er in die Punkt-Richtungsform der Mittelsenkrechten eingesetzt und somit der Mittelpunkt des Kreises berechnet.

SUB SchnittpunkteDrehachseBerechnen

  SPLinks.x = mpkreis.x - (Ebene.a * mpkreis.y) / Ebene.b

  SPLinks.y = 0

  SPLinks.z = mpkreis.z - (Ebene.c * mpkreis.y) / Ebene.b

  SPRechts.x = Ebene.a * (BildBreite - mpkreis.y) / Ebene.b + mpkreis.x

  SPRechts.y = BildBreite

  SPRechts.z = Ebene.c * (BildBreite - mpkreis.y) / Ebene.b + mpkreis.z

END SUB
Aus der Punkt-Richtungsformel der Drehachse und der Gleichung der Seitenebenen wird die Laserposition an den Seiten berechnet, damit diese Werte anschließend auszugegeben.

